

## Лабораторная работа №23

### Разработка конфигурации для учета товаров. Продажа товаров с разных складов

#### 1 Задание

Заказчик просит разработать конфигурацию для учета товаров.

Суммовой учет не ведется. Взаиморасчеты с покупателями и поставщиками не ведутся.

Учет товаров ведется в разрезе складов.

В системе необходимо регистрировать два вида операций: «Поступление товара» и «Продажа товара».

При поступлении товара пользователь в табличной части указывает, какие товары и в каком количестве поступили в организацию. Нужно предусмотреть учет до граммов.

В шапке документа выбирается склад, на который оформляется поступление. При продаже товаров указывается, какие товары были проданы и в каком количестве, с какого склада. Склад, с которого списываются товары, выбирается для каждого товара в табличной части документа.

**Продать товар «в минус» нельзя, в момент продажи следует проверять остаток товара.**

Нужно построить «Отчет» по остаткам товаров следующего вида:

Остатки товаров на 31.01.2020

Товар/Склад	Юг	Север	Запад	Итого
Ложка	100.000	40.000		140.000
Вилка	45.000		80.000	125.000
Поварешка		12.000	12.000	13.000

Отчет строится на конец дня, указанного пользователем. Особое внимание следует уделить последней секунде дня: документы, записанные на эту секунду, должны попадать в отчет.

## **2 Подготовка**

- создать новую информационную базу;
- открыть информационную базу в режиме «Конфигуратор»;
- открыть окно конфигурации.

## **3 Выполнение**

3.1 Добавить справочники «Товары», «Склады».

3.2 Открыть программу в режиме «1С:Предприятие» и добавить в каждый справочник несколько элементов.

3.3 Добавить новый документ «ПоступлениеТоваров».

3.4 Добавить реквизит «Склад», тип – СправочникСсылка.Склады, проверка заполнения – выдавать ошибку.

3.5 Добавить табличную часть «СписокТоваров», затем добавить два реквизита табличной части: «Товар» (тип – «СправочникСсылка.Товары») и «Количество» (тип – «Число», точность – 3).

3.6 Открыть программу в режиме «1С:Предприятие» и добавить в документ записи о поступлении товара.

3.7 Создать аналогичный документ «ПродажаТоваров».

3.8 На вкладке «Данные» структура должна быть немного изменена. Реквизит «Склад» следует скопировать в табличную часть. Перетаскивать реквизит нужно именно в список реквизитов табличной части.

3.9 Удалить реквизит «Склад», оставить только в табличной части.

3.10 Добавить новый регистр накопления «ОстаткиТоваровНаСкладах». Вид данного регистра – «Остатки».

3.11 Перейти на вкладку «Данные», добавить ресурс «Количество», тип – «Число», точность – 3.

Добавить измерения «Товар» (тип – «СправочникСсылка.Товары») и «Склад» (тип – «СправочникСсылка.Склады»).

3.12 Перейти на вкладку «Регистраторы», выбрать документы «ПоступлениеТоваров» и «ПродажаТоваров» в качестве документов-регистраторов.

3.13 Открыть документ «ПоступлениеТоваров», открыть окно редактирования данного документа на вкладке «Движения». Воспользоваться конструктором движений.

Поскольку получение товара должно увеличивать количество товаров на складе, то тип движения регистра необходимо выбрать «Приход». Регистр будет обозначаться знаком «+» (плюс).

Если все было сделано правильно, имена и типы реквизитов совпадают, то при нажатии на кнопку «Заполнить выражения» реквизиты регистра заполнятся автоматически. Если этого не произошло, то заполните поле «Выражение» вручную, путем выбора соответствующих реквизитов документа.

При нажатии на кнопку «ОК» система сформирует программный код, который при успешном проведении документа произведет движения в регистр накопления, то есть скопирует данные из документа в регистр накопления.

3.14 Аналогичные действия нужно проделать с документом «ПродажаТоваров».

3.15 Открыть систему в режиме «1С:Предприятие» и проверить работу регистра накопления.

В первую очередь, необходимо перепровести (провести заново) созданный документ «Поступление товаров», а также создать и провести хотя бы один документ «Продажа товаров». Без проведения документов данные не будут скопированы в регистр накопления.

Регистра накопления недостаточно для того, чтобы вести учет отрицательных остатков. Необходимо описать алгоритм работы документа «ПродажаТоваров».

3.16 Отрыть модуль объекта документа «ПродажаТоваров» и дополнить процедуру «ОбработкаПроведения».

Чтобы сделать движение данных их документа в регистр накопления, допишем после окончания цикла строку «Движения.Записать();».

Теперь, когда движение было сделано, можно обратиться к данным регистра накопления. Чтобы это сделать, воспользуемся конструктором запроса с обработкой результата. Этот конструктор можно открыть из контекстного меню щелчком правой кнопки мыши по области модуля. Данный конструктор обязательно должен быть вызван внутри процедуры «ОбработкаПроведения».

3.17 Соглашаемся с созданием нового запроса. Открывается окно конструктора запроса с обработкой результата. Перейти на вкладку «Таблицы и поля». Окно должно быть заполнено следующим образом, как показано на рисунке 1.




Рисунок 1 – Заполнение таблицы и полей

3.18 Перейти на вкладку «Условие» и добавить новое условие. Пусть в запрос попадут только данные с отрицательными остатками, как показано на рисунке 2.




Рисунок 2 – Новое условие

По окончании работы с данным конструктором нажать на кнопку «ОК». Конструктор выдаст предупреждение об ошибке, которое следует проигнорировать. Обратите внимание на текст запроса. Необходимо удалить знак амперсанта (&) перед нулем в условии.

С помощью данного запроса из базы данных можно получить отрицательные остатки по всем товарам со всех складов. Но нам нет необходимости получать такую большую выборку, нужно сузить запрос до склада, который указан в шапке документа и товаров, перечисленных в табличной части.

3.19 Открыть конструктор запроса. Для этого нужно щелкнуть в любом месте самого запроса (черный текст в двойных кавычках) правой кнопкой мыши и вызвать конструктор запроса.

3.20 Следует наложить условия на виртуальную таблицу регистра накопления. В открывшемся окне нужно написать следующее выражение:

Склад В (&МассивСкладов) И Товар В (&МассивТоваров)

Нажать на кнопку «ОК».

Данное условие поможет ограничить запрос по складам и товарам, которые указаны в табличной части документа. Нажать на кнопку «ОК». Текст запроса изменился.

3.21 Мы добавили параметры. Теперь запрос будет проводить поиск только по конкретному складу и конкретному списку товаров. Укажите склад и эти товары сразу после текста запроса.

Код процедуры полностью должен выглядеть следующим образом, как показано на рисунке 3.

```
Процедура ОбработкаПроведения(Отказ, Режим)
// регистр ОстаткиТоваровНаСкладах Расход
Движения.ОстаткиТоваровНаСкладах.Записывать = Истина;
Для Каждого ТекСтрокаСписокТоваров Из СписокТоваров Цикл
    Движение = Движения.ОстаткиТоваровНаСкладах.Добавить ();
    Движение.ВидДвижения = ВидДвиженияНакопления.Расход;
    Движение.Период = Дата;
    Движение.Товар = ТекСтрокаСписокТоваров.Товар;
    Движение.Склад = ТекСтрокаСписокТоваров.Склад;
    Движение.Количество = ТекСтрокаСписокТоваров.Количество;
КонецЦикла;
Движения.Записать ();

// контроль отрицательных остатков
Запрос = Новый Запрос;
Запрос.Текст =
"ВЫБРАТЬ
| ОстаткиТоваровНаСкладахОстатки.Товар КАК Товар,
| ОстаткиТоваровНаСкладахОстатки.КоличествоОстаток КАК КоличествоОстаток,
| ОстаткиТоваровНаСкладахОстатки.Склад КАК Склад
| ИЗ
| РегистрНакопления.ОстаткиТоваровНаСкладах.Остатки (
|
|         Склад В (&МассивСкладов)
|         И Товар В (&МассивТоваров)) КАК ОстаткиТоваровНаСкладахОстатки
| ГДЕ
| ОстаткиТоваровНаСкладахОстатки.КоличествоОстаток < 0";

Запрос.УстановитьПараметр("МассивСкладов", СписокТоваров.ВыгрузитьКолонку("Склад"));
Запрос.УстановитьПараметр("МассивТоваров", СписокТоваров.ВыгрузитьКолонку("Товар"));

РезультатЗапроса = Запрос.Выполнить();
Если НЕ РезультатЗапроса.Пустой() Тогда
    Отказ = Истина;
    Выборка = РезультатЗапроса.Выбрать();
    Пока Выборка.Следующий() Цикл
        Сообщить("На складе " + Выборка.Склад + " не хватает " +
        Выборка.КоличествоОстаток*(-1) + " шт. товаров " + Выборка.Товар);
    КонецЦикла;
КонецЕсли;
КонецПроцедуры
```

Рисунок 3 – Код

3.22 Перейти в режим «1С:Предприятие». Если у вас уже имеются документы «ПродажаТоваров», отмените их проведение. Для этого нажмите

по документу в списке правой кнопкой мыши и выберите «Отменить проведение». Сделайте это для всех документов.

3.23 Проведите документы заново. Добавьте новый документ «Продажа товаров» так, чтобы хотя бы одного из вида товаров не хватало на одном из складов. Если все было сделано правильно, и вы пытаетесь продать товаров больше, чем имеется на складе, то система выдаст ошибку.

3.24 Добавить отчет «ОтчетПоОстаткамТоваров». Воспользоваться схемой компоновки данных. В результате окна должны быть заполнены следующим образом, как показано на рисунке 4.




Рисунок 4 – Заполнение таблицы и полей

3.25 Перейти на вкладку «Ресурсы» и установить реквизит «КоличествоОстаток» в качестве ресурса: это позволит в отчете получать итоговые (просуммированные) значения.

3.26 Из условия следует, что отчет должен включать документы, записанные на последнюю секунду дня. При использовании стандартных методов такие документы в отчет попадать не будут. Поэтому нужно добавить новый параметр «ДатаОтчета» на соответствующей вкладке. Чтобы у пользователя была возможность выбирать только даты, без указания секунд, нужно настроить формат редактирования параметра «ДатаОтчета».

3.27 После нажатия кнопки «ОК» нужно настроить стандартный параметр «Период» для корректного учета последней секунды дня, как показано на рисунке 5.




Рисунок 5 – Заполнение параметров

3.28 Построить отчет в виде таблицы. Далее нужно выбрать ресурс, который будет отображен в таблице. В данном случае реквизит у нас всего один, его и выбираем. На следующем шаге следует определить те реквизиты, которыми будут заполняться колонки (Склад) и строки таблицы (Товар). Далее установить сортировку по товарам по возрастанию (то есть по алфавиту).

3.29 Чтобы у пользователя была возможность выбирать день, на который он хочет построить отчет, нужно включить параметр «ДатаОтчета» в пользовательские настройки.

3.30 Добавьте еще несколько отчетов по получению товаров и продаже товаров, чтобы убедиться, что отчет работает корректно, как показано на рисунке 6.

Параметры: Дата отчета: 08.09.2020				
Товар	Западный Количество Остаток	Северный Количество Остаток	Южный Количество Остаток	Итого Количество Остаток
Вилка	6,000	10,000	20,000	36,000
Ложка	5,000	15,000	20,000	40,000
Поварешка	4,000	17,000	20,000	41,000
Итого	15,000	42,000	60,000	117,000

Рисунок 6 – Итоговый отчет

## Лабораторная работа №24

### Разработка конфигурации для учета товаров. Контроль срока годности товаров

#### 1 Задание

Заказчик просит разработать конфигурацию для учета товаров.

Многоскладской учет не ведется. Взаиморасчеты с покупателями и поставщиками не ведутся.

В системе необходимо регистрировать поступление товара. При поступлении товара пользователь в табличной части указывает, какие товары и в каком количестве поступили в организацию. Следует предусмотреть учет до граммов. При поступлении товаров указывается срок годности партии, для каждого товара – свой.

В системе нужно регистрировать продажу товара. При продаже товаров указывается, какие товары были проданы, в каком количестве, на какую сумму. При продаже товара необходимо, в первую очередь, списывать те товары, срок годности которых подходит к концу.

*К примеру, если поставка молока «Буренка» поступила со сроками годности 30.01.2020 и 31.01.2020, то сначала списывается партия со сроком годности 30.01.2020.*

**Продать товар «в минус» нельзя, в момент продажи следует проверять остаток товара.**

Нужно построить «Отчет» по остаткам товаров следующего вида:

Остатки товаров на 15.01.2020

Товар	Срок годности	Количество
Молоко «Буренка»	28.01.2020	10.000
Молоко «Буренка»	30.01.2020	15.000
Молоко «Буренка»	31.01.2020	5.000

## **2 Подготовка**

- создать новую информационную базу;
- открыть информационную базу в режиме «Конфигуратор»;
- открыть окно конфигурации.

## **3 Выполнение**

3.1 Создать справочник «Товары».

3.2 Создать документ «ПоступлениеТоваров». Добавить табличную часть «СписокТоваров», реквизиты табличной части: «Товар» с соответствующей ссылкой на справочник, «Количество» – тип «Число», точность – 3, «СрокГодности» – тип «Дата».

3.3 Добавить регистр накопления «ОстаткиТоваров» вида «Остатки». Добавить ресурс – «Количество», тип «Число», точность – 3. Измерения: «Товар» (тип – «СправочникСсылка.Товары») и «Срок годности» (тип – «Дата»).

3.4 Открыть окно редактирования документа «ПоступлениеТоваров» на вкладке «Движения». Выбрать регистр, в который будут передаваться данные документа, и воспользоваться конструктором движений.

Поскольку получение товара должно увеличивать общее количество товаров, то тип движения регистра необходимо выбрать «Приход». Регистр будет обозначаться знаком «+» (плюс).

3.5 Добавить отчет «ОстаткиТоваров».

Создавать отчет будем с помощью формы. Переходим на вкладку «Формы» и создаем новую форму отчета. Добавить новый реквизит «ТабДок», тип – «Табличный документ». Перенести созданный реквизит на форму.

В области предпросмотра должна появиться заготовка табличного документа.

3.6 Далее добавим кнопку, при нажатии на которую будет формироваться отчет, как показано на рисунке 2.




Рисунок 1 – Создание кнопки

Дадим кнопке имя «Сформировать», перенесем ее на форму, а затем опишем ее действие, как показано на рисунке 2.




Рисунок 2 – Создание кнопки

Внутри процедуры «СформироватьНаСервере» откроем конструктор запроса с обработкой результата. Для этого нужно вызвать контекстное меню правой кнопкой мыши, щелкнув по пустой области внутри процедуры. Из контекстного меню выбрать необходимый конструктор. Создадим новый запрос.

Откроется окно конструктора запроса. На вкладке «Обработка результата» из предложенных вариантов выбора обработки необходимо выбрать вариант вывода в табличный документ.

3.7 Перейти на следующую вкладку «Таблицы и поля». Результат заполнения представлен на рисунке 3.




Рисунок 3 – Заполнение таблицы и полей

3.8 Далее переключиться на вкладку «Порядок». Здесь можно упорядочить элементы так, как нам удобно. Упорядочим их сначала по наименованию, а затем – по сроку годности.

3.9 Далее переходим на вкладку «Итоги» и просуммируем все товары между собой. Для этого нужно заполнить поля следующим образом, как показано на рисунке 4.




Рисунок 4 – Заполнение полей

По завершении работы с конструктором движений нажимаем на кнопку «ОК». Система на основе введенных ранее настроек сформировала макет, по которому будет строиться отчет.

Внутри процедуры «СформироватьНаСервере» должен появиться код, который будет заполнять макет данными.

3.10 Теперь попробуем запустить систему в режиме «1С:Предприятие» и проверить, корректно ли работают созданные нами объекты.

3.11 Теперь добавим несколько документов «Поступление товаров», причем для одних и тех же продуктов укажем разные сроки годности.

3.12 А теперь попробуем сформировать отчет. Откроем его и нажмем на кнопку «Сформировать», которая находится под таблицей. Получаем отчет, как показано на рисунке 5.

Товар	СрокГодности	КоличествоОстаток
Молоко "Буренка"		35,000
	17.09.2020	
	0:00:00	20,000
Творог "Элитный"	19.09.2020	
	0:00:00	15,000
		28,000
	19.09.2020	
	0:00:00	18,000
	21.09.2020	
	0:00:00	10,000

Рисунок 5 – Отчет

В отчете мы можем увидеть общее количество тех или иных товаров с различными сроками годности (обозначено полужирным). Произведено упорядочивание товаров, в первую очередь, по алфавиту, а затем уже – по сроку годности.

Данный отчет понадобится нам для реализации продажи товаров. Сначала мы можем посмотреть, сколько всего единиц того или иного товара у нас имеется, и только затем уже списывать товары из различных партий (с различным сроком годности).

3.13 Аналогично документу «ПоступлениеТовара» создайте документ «ПродажаТовара». Обратите внимание, что срок годности учитывать не нужно.

3.14 Теперь следует описать логику проведения данного документа. Для этого нужно открыть модуль объекта, который находится на вкладке «Прочее».

В документ «ПродажаТоваров» пользователь должен добавлять информацию о товаре и количестве, которое он хочет продать. В первую очередь, система должна проанализировать, достаточно ли каждого вида товаров для продажи. Если достаточно, тогда нужно списывать товар партиями по возрастанию срока годности.

Для проведения данного анализа необходимо получить данные как из документа, так и из регистра накопления. Получим эти данные с помощью конструктора запросов.

Этот конструктор можно открыть из контекстного меню щелчком правой кнопки мыши по области модуля. Данный конструктор обязательно должен быть вызван внутри процедуры «ОбработкаПроведения».

Открывается окно конструктора запроса с обработкой результата. Переходим на вкладку «Таблицы и поля». Здесь нам нужно выбрать поля табличной части документа, как показано на рисунке 6.




Рисунок 6 – Заполнение таблицы и полей

Переходим на вкладку «Группировка» и сгруппируем товары между собой, как показано на рисунке 7: вдруг пользователь ввел несколько отдельных строк с одинаковым товаром?




Рисунок 7 – Группировка

Далее переходим на вкладку «Условие» и поставим условие на ссылку. Таким образом, запрос будет обрабатывать табличную часть только того одного документа, чью ссылку мы передадим в запрос, как показано на рисунке 8.




Рисунок 8 – Заполнение условия

3.15 Теперь все полученные данные из документа «ПродажаТоваров» следует поместить во временную таблицу. Для этого переходим на вкладку

«Дополнительно», выбираем пункт «Создание временной таблицы» и зададим ей имя «ДОК».

3.16 Переходим на вкладку «Индекс» и проиндексируем таблицу по полю «Товар», поскольку далее нужно будет соединить несколько таблиц.

3.17 Переходим на вкладку «Пакет запросов» и добавляем новый запрос.

Снова открывается вкладка «Таблицы поля». Здесь нужно выбрать уже остатки из регистра накопления, как показано на рисунке 9.




Рисунок 9 – Заполнение таблицы и полей

3.18 Полученные данные тоже положим во временную таблицу. Переходим на вкладку «Дополнительно» и создаем временную таблицу «ОСТ». Переходим на вкладку «Индекс» и снова индексируем по товару. Осталось лишь добавить последний запрос, который сможет объединить в себе данные из двух получившихся временных таблиц. Вкладка «Пакет запросов», кнопка «Добавить». Необходимо выбрать следующие поля из двух таблиц, как показано на рисунке 10.




Рисунок 10 – Заполнение таблицы и запроса

3.19 Теперь нужно перейти на вкладку «Связь» и связать между собой две временные таблицы по полю «Товар», как показано на рисунке 11.




Рисунок 11 – Установление связи

3.20 Далее переходим на вкладку «Порядок» и упорядочиваем элементы в запросе по товару и сроку годности (по возрастанию).

3.21 Далее переходим на вкладку «Итоги» и снова просуммируем все товары, как показано на рисунке 12.

Рисунок 12 – Заполнение итогов

По окончании работы над итогами нажимаем на кнопку «ОК». В процедуре «ОбработкаПроведения» появился код, который был сформирован конструктором запроса. Под текстом запроса можно увидеть цикл в цикле: внешний цикл перебирает товары, а внутренний – сроки годности этих товаров.

*Пример: Предположим, что некий клиент решил купить 50 пакетов молока. В первую очередь, нужно проверить, а есть ли у нас 50 пакетов на складе?*

Именно такую проверку и нужно добавить. В случае нехватки товаров нужно вывести пользователю сообщение и отметить проведение, как показано на рисунке 13.

```
Пока ВыборкаТовар.Следующий() Цикл
    // сколько хотят купить // сколько осталось на складе
    Если ВыборкаТовар.Количество > ВыборкаТовар.КоличествоОстаток Тогда
        Сообщить ("Не хватает товаров " + ВыборкаТовар.Товар + " " +
            (ВыборкаТовар.Количество - ВыборкаТовар.КоличествоОстаток) + " шт.");
        Отказ = Истина;
    КонецЕсли;
    ВыборкаДетальныеЗаписи = ВыборкаТовар.Выбрать ();

    Пока ВыборкаДетальныеЗаписи.Следующий() Цикл
        // Вставить обработку выборки ВыборкаДетальныеЗаписи
    КонецЦикла;
КонецЦикла;
```

Рисунок 13 – Код

Теперь будем перебирать партии.

*Пример* Предположим, что в документе указано 50 пакетов молока. На складе в одной партии – 20 пакетов, а в другой – 40. Необходимо сначала списать 20 из первой партии, а затем – 30 из второй.

Добавим переменную «ОсталосьСписать» – она будет своеобразным счетчиком, который мы будем вести до нуля, как показано на рисунке 14.

```
Пока ВыборкаТовар.Следующий() Цикл
// сколько хотят купить //сколько осталось на складе
Если ВыборкаТовар.Количество > ВыборкаТовар.КоличествоОстаток Тогда
    Сообщить ("Не хватает товаров " + ВыборкаТовар.Товар + " " +
        (ВыборкаТовар.Количество - ВыборкаТовар.КоличествоОстаток) + " шт.");
    Отказ = Истина;
КонецЕсли;
ОсталосьСписать = ВыборкаТовар.Количество;
ВыборкаДетальныеЗаписи = ВыборкаТовар.Выбрать();
Пока ВыборкаДетальныеЗаписи.Следующий() Цикл
    // вставить обработку выборки ВыборкаДетальныеЗаписи
КонецЦикла;
КонецЦикла;
```

Рисунок 14 – Код

Теперь проверяем, сколько осталось товаров в конкретной партии и списывать это количество из переменной «ОсталосьСписать», после чего будем переходить к следующей партии. Списывать из конкретной партии можно не больше, чем в ней есть, или меньше. Воспользуемся функцией «Мин», которая выбирает наименьшее из заданных в нее чисел, как показано на рисунке 15.

```
Пока ВыборкаТовар.Следующий() Цикл
// сколько хотят купить //сколько осталось на складе
Если ВыборкаТовар.Количество > ВыборкаТовар.КоличествоОстаток Тогда
    Сообщить ("Не хватает товаров " + ВыборкаТовар.Товар + " " +
        (ВыборкаТовар.Количество - ВыборкаТовар.КоличествоОстаток) + " шт.");
    Отказ = Истина;
КонецЕсли;
ОсталосьСписать = ВыборкаТовар.Количество;
ВыборкаДетальныеЗаписи = ВыборкаТовар.Выбрать();
Пока ВыборкаДетальныеЗаписи.Следующий() Цикл
    Списать = Мин(ОсталосьСписать, ВыборкаДетальныеЗаписи.КоличествоОстаток);
КонецЦикла;
КонецЦикла;
```

Рисунок 15 – Код

Далее нужно уменьшить количество товаров, которые еще нужно списать, и добавить дополнительное условие на переменную «ОсталосьСписать». Установим маркер формирования движений в значение «Истина». Проверить получившийся код, как показано на рисунке 16.

```

Процедура ОбработкаПроведения(Отказ, Режим)
Запрос = Новый Запрос;
Запрос.Текст =
"ВЫБРАТЬ
|     ПродажаТоваровСписокТоваров.Товар КАК Товар,
|     СУММА(ПродажаТоваровСписокТоваров.Количество) КАК Количество
|ПОМЕСТИТЬ ДОК
|ИЗ
|     Документ.ПродажаТоваров.СписокТоваров КАК ПродажаТоваровСписокТоваров
|ГДЕ
|     ПродажаТоваровСписокТоваров.Ссылка = &Ссылка
|
|СГРУППИРОВАТЬ ПО
|     ПродажаТоваровСписокТоваров.Товар
|
|ИНДЕКСИРОВАТЬ ПО
|     Товар
|;

////////////////////////////////////
|ВЫБРАТЬ
|     ОстаткиТоваровОстатки.Товар КАК Товар,
|     ОстаткиТоваровОстатки.СрокГодности КАК СрокГодности,
|     ОстаткиТоваровОстатки.КоличествоОстаток КАК КоличествоОстаток
|ПОМЕСТИТЬ ОСТ
|ИЗ
|     РегистрНакопления.ОстаткиТоваров.Остатки КАК ОстаткиТоваровОстатки
|
|ИНДЕКСИРОВАТЬ ПО
|     Товар
|;

////////////////////////////////////
|ВЫБРАТЬ
|     ДОК.Товар КАК Товар,
|     ДОК.Количество КАК Количество,
|     ОСТ.СрокГодности КАК СрокГодности,
|     ОСТ.КоличествоОстаток КАК КоличествоОстаток

|ИЗ
|     ДОК КАК ДОК
|         ЛЕВОЕ СОЕДИНЕНИЕ ОСТ КАК ОСТ
|         ПО ДОК.Товар = ОСТ.Товар
|
|УПОРЯДОЧИТЬ ПО
|     Товар,
|     СрокГодности
|ИТОГИ
|     МИНИМУМ(Количество),
|     СУММА(КоличествоОстаток)
|ПО
|     Товар";

Запрос.УстановитьПараметр("Ссылка", Ссылка);

РезультатЗапроса = Запрос.Выполнить();

```

```

ВыборкаТовар = РезультатЗапроса.Выбрать (ОбходРезультатаЗапроса.ПоГруппировкам);

Пока ВыборкаТовар.Следующий() Цикл
    // сколько хотят купить // сколько осталось на складе
    Если ВыборкаТовар.Количество > ВыборкаТовар.КоличествоОстаток Тогда
        Сообщить ("Не хватает товаров " + ВыборкаТовар.Товар + " " +
            (ВыборкаТовар.Количество - ВыборкаТовар.КоличествоОстаток) + " шт.");
        Отказ = Истина;
    КонецЕсли;
    ОсталосьСписать = ВыборкаТовар.Количество;
    ВыборкаДетальныеЗаписи = ВыборкаТовар.Выбрать ();
    Пока ВыборкаДетальныеЗаписи.Следующий() И ОсталосьСписать > 0 Цикл
        Списать = Мин (ОсталосьСписать, ВыборкаДетальныеЗаписи.КоличествоОстаток);
        Движение = Движения.ОстаткиТоваров.Добавить ();
        Движение.ВидДвижения = ВидДвиженияНакопления.Расход;
        Движение.Период = Дата;
        Движение.Товар = ВыборкаДетальныеЗаписи.Товар;
        Движение.СрокГодности = ВыборкаДетальныеЗаписи.СрокГодности;
        Движение.Количество = Списать;
        ОсталосьСписать = ОсталосьСписать - Списать;
    КонецЦикла;
КонецЦикла;
Движения.ОстаткиТоваров.Записывать = Истина;

КонецПроцедуры

```

Рисунок 16 – Код

3.22 Проверим работоспособность разработанной информационной системы. Откроем отчет по остаткам, как показано на рисунке 17.

Товар	СрокГодности	КоличествоОстаток
Молоко "Буренка"		<b>35,000</b>
	17.09.2020	
	0:00:00	20,000
	19.09.2020	
Творог "Элитный"	0:00:00	15,000
		<b>28,000</b>
	19.09.2020	
	0:00:00	18,000
	21.09.2020	
	0:00:00	10,000

Рисунок 17 – Отчет

Всего имеется 35 единиц молока «Буренка». Попробуем продать 25 единиц молока. Для этого добавим документ «Продажа товаров». Полученный документ необходимо провести, как показано на рисунке 18.

Продажа товаров (создание) \*

Провести и закрыть Записать Провести Еще ▾

Номер:

Дата: 10.09.2020 0:00:00

Добавить Поиск (Ctrl+F) × Еще ▾

N	Товар	Количество
1	Молоко "Буренка"	25,000

Рисунок 18 – Продажа товаров

Если появились проблемы с проведением документа «Продажа товаров», тогда вернитесь к списку всех документов и отмените их проведение. Для этого выделите документ в списке, нажмите на него правой кнопкой мыши и выберите пункт меню «Отменить проведение». Сделайте это для всех документов. Затем проведите документы заново.

Вернемся к отчету и сформируем его, как показано на рисунке 19.

Товар	СрокГодности	КоличествоОстаток
Молоко "Буренка"	19.09.2020	10,000
	0:00:00	10,000
Творог "Элитный"	19.09.2020	28,000
	0:00:00	18,000
	21.09.2020	10,000
	0:00:00	

Рисунок 19 – Отчет

Система списала 20 единиц молока из партии со сроком годности до 17 сентября, и еще 5 – из партии со сроком годности до 19 сентября.

## Самостоятельная работа (обязательно к выполнению)

### Задание

Заказчик просит разработать конфигурацию для учета товаров аптеки (лекарственных средств).

Учет ведется исключительно в количественном выражении (штуки, граммы). Взаиморасчеты с покупателями и поставщиками не ведутся.

Учет товаров ведется в разрезе складов (аптечных пунктов). Для каждой приходуемой партии обязательно указывается ее срок годности.

В системе необходимо регистрировать два вида операций: «Поступление товара» и «Реализация товара».

При поступлении товара пользователь выбирает склад-получатель. В табличной части документа указывает, какие товары, в каком количестве и с каким сроком годности поступили.

При реализации товаров пользователь в шапке документа указывает склад-источник. В табличной части выбирает товары и количество для продажи. Для каждой строки можно указать другой склад списания, если товар берется не с основного. В момент проведения документа система должна автоматически списать товары с выбранных складов из партий с наименьшим сроком годности. Реализовать товар при отсутствии достаточного остатка на складе невозможно.

Необходимо учитывать, что документы могут вводиться задним числом. При этом данные по остаткам и последовательности списания партий должны оставаться корректными.

В результате выполнения работы необходимо создать отчет следующего вида:

Остатки товаров на складах на 31.01.2020

Склад	Номеклатура	Срок годности	Количество
....	.....	.....	....